

Bangladesh University of Engineering and Technology  
Electric Circuits II Laboratory (EEE 106)  
Simulation Experiment No. 02  
**Steady-State AC analysis, Frequency response and Filters**

## Pre Lab

- Read the Lab Sheet.
- Go to Figure 1. Use pen and paper to calculate the Voltages and Currents of all the nodes at 60Hz. Also, draw the phasor diagram.
- Go to Figure 2. Use pen and paper to find the expression of voltage (magnitude and phase) at node 2 in terms of frequency ( $f$ , where  $f = 2 \times \pi \times f$ ).
- Determine the cut off frequency of the filter shown in Figure 2: equate the expression of the voltage at node 2 to  $1/\sqrt{2}$  and find the corresponding frequency.

## Introduction

In addition to DC circuit analysis, DC transient analysis, and AC transient analysis PSpice can be used to work on AC steady-state analysis.

## AC Voltage and Current Sources

The syntax for an AC source is very similar to its DC counterpart. The AC source is assumed to be a *cosine* waveform at a specified phase angle. Its frequency must be defined in a separate ".AC" command that defines the frequency for *all* the sources in the circuit. The unique information for the individual source is: the name, which must start with "V" or "I," the node numbers, the magnitude of the source, and its phase angle. Some examples follow.

*name	nodelist		type	rms value	phase(deg)
Vac	4	1	AC	120V	30
Vba	2	5	AC	240	; phase angle 0 degree
Ix	3	6	AC	10.0A	-45 ; phase angle -45 degree
Isv	12	9	AC	25mA	; 25 milliamps @ 0 degree

Here source type, AC, *must* be specified, because the default is DC. If the phase angle is not specified it will be assumed as zero degrees. The units of the phase angle will be in degrees. For voltage source the node on the left is the positive node and the node on the right is the negative node. Similarly, in case of current source positive current flows into the source from the node on the left, passes through the source, and leaves the source from the node on the right.

In some of your previous experiments you have used "SIN" type of source which is one of several useful source types (also EXP, PULSE, PWL, etc) that are used for *transient analysis*. Do not attempt to use SIN for steady-state (phasor) AC analysis nor for frequency sweeps. The

SIN type is a time-based function for time-based analysis, whereas the AC type is used in frequency-based modeling. Since phasor analysis uses frequency-based models of circuit elements, always use the AC type as described in this experiment.

## Use of the .PRINT AC Command

To enable **.PRINT** command **.AC** command must be used. The **.AC** command was designed to make a sweep of many frequencies for a given circuit. This is called a *frequency response*. Three types of ranges are possible for the frequency sweep: LIN, DEC and OCT. At this time we only want a single frequency to be used so it does not matter which one we choose. We will pick the LIN (linear) range to designate our single frequency.

```
*   type #points  start stop
.AC  LIN    1      60Hz  60Hz; <== single frequency
.AC  LIN    6      100   200; <== a linear range sweep
.AC  DEC   20      1Hz   10kHz; <== a logarithmic range sweep
```

The first statement above performs a single analysis using the frequency of 60 Hz. Placing the units "Hz" after the value is optional. The second statement would perform a frequency sweep using frequencies of 100Hz, 120Hz, 140Hz, 160Hz, 180Hz, and 200Hz. The third statement performs a logarithmic range sweep using 20 points per decade over a range of four decades. This will be useful later for studying frequency response of circuits.

Finally, we can discuss the actual **.PRINT AC** command. Printing the components of phasor values (complex numbers) requires some options. There are four expressions needed for this: magnitude, phase (angle), real part and imaginary part. In addition, we can print voltages or currents. For instance, to print the magnitude of a voltage between nodes 2 and 3, we would specify "VM(2,3)." The phase angle of this same voltage would be "VP(2,3)" and would be printed in degrees. If we need the current magnitude through resistor Rload, we would specify "IM(Rload)." The real part of the voltage on node 7 would be specified "VR(7)" and its imaginary part, "VI(7)." As with the **.PRINT DC** command, there is no limit on the number of times it can be used in a listing; nor is there a limit on how many print requests can be on a single line.

```
.PRINT AC VM(30,9) VP(30,9); magnitude & angle of voltage
.PRINT AC IR(Rx) II(Rx); real & imag. parts of current through Rx
.PRINT AC VM(17) VP(17) VR(17) VI(17); the whole works on node 17
```

## Specifying frequency range for AC Sources

.AC command is used to specify one of the following three types of frequency ranges.

### *LIN (linear) Range Type*

The LIN range type is linear. It divides up the range between the minimum and maximum user-specified frequencies into evenly spaced intervals. This is best used to view details over a narrow bandwidth. The first parameter after the keyword LIN is the number of points to calculate. This is followed by the lowest frequency value in Hz, then the highest frequency value in Hz. As with all the range types, the unit "Hz" is optional.

```
*   type #points  start stop
.AC LIN  101      2k    4k    ; 101 points from 2 kHz to 4 kHz
.AC LIN  11       800   1000 ; 11  points from 800 Hz to 1 kHz
```

### *OCT (octal) Range Type*

The OCT range is logarithmic to the base two. Thus each octave has the same number of points calculated. This is somewhat useful for designing electronic equipment for musical applications. However, the resulting graphs are very similar in appearance to sweeps made with the DEC range. The first parameter after the keyword OCT is the number of points per octave to calculate. This is followed by the lowest frequency value in Hz, then the highest frequency value in Hz.

```
*   type #points  start stop
.AC OCT  20 440Hz 1.76kHz; 20 points/octave over 2 octaves
.AC OCT  40 110Hz 880Hz  ; 40 points/octave over 3 octaves
```

### *DEC (decade) Range Type*

The DEC range is logarithmic to the base ten. Thus each decade has the same number of points calculated. This is the most commonly used range for making *Bode plots* of a frequency response. The first parameter after the keyword DEC is the number of points per decade to calculate. This is followed by the lowest frequency value in Hz, then the highest frequency value in Hz.

```
*   type #points  start stop
.AC DEC  50      1kHz 100kHz ; 50 points/decade over 2
                        ; decades
.AC DEC  25      100k 100MEG ; 25 points/decade over 3
                        ; decades
```

## Probe

The independent variable used by PROBE in a .TRAN analysis is *time*. But in a frequency sweep the independent variable used by PROBE is *frequency*. When PROBE stores data in a transient (.TRAN) analysis, the dependent variables are instantaneous voltages and currents; whereas in a frequency sweep these dependent variables are real and imaginary components of phasor voltages and currents.

## Example Circuit 1: Steady State AC Analysis

We will analyze the following circuit at a frequency at 60 Hz.




Figure 1. Circuit for Example 1

### Corresponding Netlist of the Circuit

```
;60 Hz AC Circuit
Vs 1 0 AC 120V 0
Rg 1 2 0.5
Lg 2 3 3.183mH
Rm 3 4 16.0
Lm 4 0 31.83mH
Cx 3 0 132.8uF
.AC LIN 1 60 60
.PRINT AC VM(3) VP(3)
.PRINT AC IM(Rm) IP(Rm)
.PRINT AC IM(Cx) IP(Cx)
.END
```

## Example Circuit 2: Frequency Sweep

We will look in to a simple circuit to understand AC Frequency Sweep.




Figure 2: Circuit for Example 2

Corresponding Netlist of the Circuit

```
;First-order low-pass RC filter
Vin 1 0 AC 1.0V
R1 1 2 0.25
C1 2 0 50uF
.AC DEC 20 100Hz 100kHz
.PROBE
.END
```

The circuit in Fig. 2 is a first-order low-pass filter. Since we want the gain of this filter, it is convenient to make the input voltage 1 volt so the output voltage is numerically equivalent to the gain. However, the post-processor within PROBE is fully capable of performing arithmetic such as dividing the input voltage into the output voltage.

### Plotting In Terms of Voltage

After running this in PSpice, start PROBE, choose "Add" from the "Trace" menu and plot the output voltage. PROBE will provide the following graph.




Figure 3: Voltage across Capacitor for the Circuit in Fig. 2.

### Plotting the Gain (in Terms of Decibel)

Another option is to have PROBE plot the gain in decibels. To do this, choose "Add" from the "Trace" menu in PROBE. Then select the "DB" function in the right-hand column and choose "V(2)" from the left-hand column. After selecting "OK," you should see the following trace.




Figure 4: Plot in terms of Decibel

Notice that the gain is -3dB at a frequency of 1 kHz (the half-power frequency) and declines at 20 dB/decade thereafter.

### Plotting the Phase

The remaining demonstration for this example is to have PROBE plot the *phase shift* of the low-pass filter as a function of frequency. We simply specify "VP(2)" from the "Add Trace" dialog box. Notice that this is the same format used in the .PRINT AC command in PSPICE. PROBE automatically shows the angles in degrees.




Figure 5: Plot of the Phase


## Post Lab

Answer the following questions in your lab report. Your report should include necessary figures, plots and explanations.

1. In the example 2, we used a source voltage of  $1\angle 0$ . And plotted the gain in decibel just by plotting the voltage at node 2. But in real life the source voltage would not necessary be unity. It may as well be any arbitrary voltage  $V\angle\theta$ . How would you plot the gain (in decibels) in such case? Devise a method.

Use  $6\angle 0$  in the circuit of figure 2. Perform frequency Sweep and Plot the gain by the method you devised. At what decibel do you get the cutoff frequency this time? Does the cutoff frequency depend on source voltage?


2. Determine the cutoff frequency of the high-pass filter shown in figure 6 by pen and paper. Simulate the circuit using Frequency Sweep. Keep the cutoff frequency within the range of the sweep. Find the cutoff frequency from the plot of the voltage. Does the frequency obtained from the simulation and from the pen and paper match?



**Figure 6: High Pass**

3. Cascade the low-pass and high-pass filters shown in figure 2 and figure 6 and form a band pass filter.

Then find the center frequency of the circuit using frequency sweep. Center frequency is the frequency at which the gain is highest.



**Figure 7: Band Pass Filter.**

And then find the lower and upper cutoff frequencies. These frequencies are the frequencies at which the gain is -3db.

4. Now, we will explore the second order filters using RLC circuits. The following circuits are collected from the Wikipedia page of the RLC Circuits ([https://en.wikipedia.org/wiki/RLC\\_circuit](https://en.wikipedia.org/wiki/RLC_circuit)). You can look up in the page and find out what kind of filters are each of these. However, it will be beneficial if you perform the frequency sweep, look at the gain plot and determine what kind of circuits are these.

As usual use  $1\angle 0$  as the source voltage for each of these circuits. Perform, frequency sweep and determine what kind of filters are these. We are not necessary interested at the cutoff frequencies for the purpose of the report. Look up the Wikipedia page and see the equations of the cutoff frequencies. These are easy to derive following standard procedure.




Figure 8 (a)




Figure 8 (b)




Figure 8 (c)




Figure 8 (d)




Figure 8 (e)




Figure 8 (f)

Use suitable values for  $R_L$ ,  $L$ , and  $C$  for simulation.

5. There are 8 configurations in 4. There are few more configurations which can be designed. Can you design your very own second order filter? Perform a frequency sweep on the filter you designed. What kind of filter it is? What are the cut off frequency/frequencies and center frequency (if any)?

(Optional: Find the expression of the output voltage for the filter you designed and comment on the characteristics of the filter based on the equation.)